PharmaSUG 2015 – Paper #PO23

Analysis Methods for a Sequential Parallel Comparison Design (SPCD)

Harry Haber, MMS Holdings, Inc., Canton, MI, USA
James Zuazo, MMS Holdings, Inc., Canton, MI, USA
Linda LaMoreaux, MMS Holdings, Inc., Canton, MI, USA

ABSTRACT

There are many considerations to take into account when planning a clinical trial, and an important consideration is trial design. Adaptive designs are sometimes chosen, as they include features that will increase the efficiency of a trial. One such example is the Sequential Parallel Comparison Design (SPCD), an adaptive design that allows for re-randomization of specific placebo subjects ("non-responders") from an early stage to placebo or treatment in a subsequent stage of the trial. This increases the number of subjects who receive active treatment and improves the power of the trial. Trials using the SPCD design can be analyzed using a number of different methods, including Ordinary Least Squares (OLS), Seemingly Unrelated Regression (SUR) and Repeated Measures Linear Model (MMRM). These methods are described in three papers that feature the SPCD design. This paper will describe, display and compare the methods. The SPCD design is currently in its early stages of implementation and acceptance. The efficiencies of this design will ensure an increase in popularity and a realization of the benefits through its application. This paper will give guidance on how to analyze studies that use this design.

INTRODUCTION

Adaptive designs are sometimes used in clinical trials in order to increase the efficiency of the trial. The Sequential Parallel Comparison Design (SPCD) is a patented adaptive design that allows for re-randomization of placebo non-responders from an early stage to placebo or treatment in a subsequent stage of the trial. This clinical trial design was developed at Massachusetts General Hospital (MGH) by Dr. Maurizio Fava[1], and Dr. David Schoenfeld.

In placebo-controlled trails, subjects are given either active drug or placebo treatment in blinded fashion. Subjects on the placebo arm will sometimes display a perceived or actual response. High placebo response can cause a failure to show a difference between drug and placebo, resulting in a negative trial. SPCD reduces the detrimental impact of placebo response by including two stages of treatment and utilizing each subject at least once (sometimes twice) for efficacy analysis. In comparison to a conventional trial design with a specified sample size, the use of SPCD with the same sample size can increase the power of a trial, by approximately 10-25 percentage points. Alternatively, if a particular power is specified, SPCD can allow a sample size reduction by 20% - 50%.

DESCRIPTION OF SPCD

SPCD is a flexible design which can involve a number of possible formats. The SPCD trial design considered here involves two double-blind stages of treatment, with re-randomization to Stage 2 treatment occurring immediately at the conclusion of Stage 1. Neither subjects nor clinicians are aware of when Stage 1 ends and Stage 2 begins. The efficacy analysis is based on all data from Stage 1 and the Stage 2 data from subjects who were placebo non-responders in Stage 1 and re-randomized. Therefore, data from both stages are utilized for the efficacy analysis, with response data of all subjects utilized at least once, and response data of some subjects utilized twice.

SPCD can be used for both binary and continuous data and allows for a number of different analysis methods. The following analysis methods for SPCD trials have been described in the literature:

1. Ordinary Least Squares (OLS), (Chen[2] et al, 2011)
2. Seemingly Unrelated Regression (SUR), (Tamura[3] and Huang, 2007)
3. Repeated Measures Linear Model (MMRM), (Doros[4], 2013).

This paper will describe how to implement these methods using SAS code. A sample dataset will be provided for illustration and details will be given on how to use the SAS output to construct the test statistic in each case.

EXAMPLE CLINICAL TRIAL AND SAMPLE DATA

For our example, a 10-week randomized, double-blind, placebo-controlled study using the Sequential Parallel Comparison Design (SPCD) will be considered. This study consists of 2 consecutive double-blind treatment stages (Stage 1 and Stage 2), with a planned re-randomization at the end of Stage 1 (i.e., Week 5). The efficacy outcome measure is a continuous variable. Observations at Week 5 are used both as Stage 1 assessment and baseline for Stage 2.

The schematic below illustrates in the design used in the example trial. The letters A through G identify the SPCD subject groupings. At Stage 1, subjects are randomized to either Placebo (A) or Drug (B). Subjects re-randomized at Stage 2 will be placed into one of the following groupings:

1. Subjects on Drug at Stage 1 (B) will continue on drug (G),
2. Subjects on Placebo at Stage 1 (A), will be classified as either non-responders or responders.
    a. Non-responder subjects will be re-randomized to either Placebo (C) or Drug (D).
    b. Responder subjects will be re-randomized to either Placebo (E) or Drug (F).



Figure 1. Study Schematic

Sample data for 22 subjects from the example trial are given below:

SUBJID	BASE	WEEK5	WEEK10	CHG1	CHG2	TRT01PN	TRT02PN	RESPFL
1	6	0	0	-6	0	1	1	Y
2	2	0	0	-2	0	0	1	Y
3	4	0	6	-4	6	0	1	Y
4	6	4	0	-2	-4	0	1	N
5	4	0	1	-4	1	0	0	Y
6	6	0	0	-6	0	0	1	Y
7	8	8		0		1		
8	6	6	2	0	-4	0	1	N
9	6	12	4	6	-8	0	1	N
10	6	6		0		1		
11	6	6		0		0		
12	6	2	1	-4	-1	1	1	Y
13	6	0		-6		1		
14	8	2	2	-6	0	1	1	N
15	6	9	9	3	0	0	0	N
16	8	8	8	0	0	1	1	N
17	12	8		-4		1		
18	6	6	6	0	0	0	0	N
19	6	6	4	0	-2	0	0	N
20	4	4	4	0	0	0	1	N
21	6	6	6	0	0	1	1	N
22	6	6	6	0	0	1	1	N

Table 1. Sample Data From Example Clinical Trial

The variables in the sample dataset are as follows:

- SUBJID: Subject identification number
- BASE: Efficacy outcome measure at Stage 1 baseline
- WEEK5: Efficacy outcome measure at Week 5, i.e. end of Stage 1 (also serves as baseline for Stage 2)
- WEEK10: Efficacy outcome measure at Week 10, i.e. end of Stage 2
- CHG1: Change from Stage 1 baseline to Week 5
- CHG2: Change from Stage 2 baseline (i.e. Week 5) to Week 10
- TRT01PN: Randomized treatment in Stage 1 (0=placebo, 1=Drug)
- TRT01PN: Randomized treatment in Stage 2 (0=placebo, 1=Drug)
- RESPFL: Responder status at end of Stage 1 (Y=yes, N=no, blank=did not enter Stage 2).

The three analysis methods (OLS, SUR and MMRM) will be described below.

OLS METHOD

For the OLS method, Stage 1 (A vs. B from Figure 1) and Stage 2 (placebo non-responders only – C vs.D from Figure 1) are analyzed separately and the weighted treatment effects from the stages are combined to give an overall treatment effect. SAS PROC REG was applied using data from each Stage separately to test the difference in change from baseline between treatments. The Stage 1 analysis included all subjects randomized to placebo or Drug and the Stage 2 analysis included only data from subjects randomized to placebo at Stage 1 who were non-responders. Stage 2 data are ignored for placebo responders and for subjects who were randomized to Drug at Stage 1.

SAS code for this method is given below:

```
*********************************************************;
*********************** OLS Method **********************;
*********************************************************;

/*** Stage 1 OLS model                                   ***/
/*** Subjects meeting all of the below criteria will be ***/
/***   included in the stage 1 model:                    ***/
/***   1)Planned Stage 1 treatment                       ***/
/***   2)Stage 1 change from baseline                    ***/
ods output parameterestimates=olsreg1;
ods graphics off;
      proc reg data=SPCD;
            model CHG=BASE TRT01PN;
      run;
quit;

/*** Stage 2 OLS model                                   ***/
/*** Subjects meeting all of the below criteria will be ***/
/***   included in the stage 2 model:                    ***/
/***   1)Placebo Stage 1 non-responder                   ***/
/***   2)Planned Stage 2 treatment                       ***/
/***   3)Stage 2 change from baseline                    ***/
ods output parameterestimates=olsreg2;
      proc reg data=SPCD (where=(TRT01PN=0 and RESPFL="N"));
            model CHG2=BASE2 TRT02PN;
      run;
quit;
```

The output produced by this code for our sample data is as follows:

Stage 1 OLS Model Results

Parameter Estimates						
Variable	Label	DF	Parameter Estimate	Standard Error	t Value	Pr > |t|
Intercept	Intercept	1	-1.88486	2.37964	-0.79	0.4381
BASE	Baseline Value	1	0.21965	0.42648	0.52	0.6125
TRT01PN	Planned Treatment for Period 01 (N)	1	-2.29662	1.59003	-1.44	0.1649

Stage 2 OLS Model Results

Parameter Estimates						
Variable	Label	DF	Parameter Estimate	Standard Error	t Value	Pr > |t|
Intercept	Intercept	1	3.61905	2.33090	1.55	0.1955
BASE2	Analysis Value at Week 5	1	-0.61224	0.28837	-2.12	0.1010
TRT02PN	Planned Treatment for Period 02 (N)	1	-3.63946	1.54847	-2.35	0.0785

For each model, the parameter estimates come from the PROC REG parameterestimates dataset and will be used to calculate OLS weighted statistic. The OLS weighted statistic will combine the two Stage estimates into a single weighted Z-statistic. For our example, we will weight each stage evenly giving a weight of .5 to each of the stage results. Following the below formula, the weighted Z OLS statistic is calculated:

$$Z_{OLS} = \frac{weight(stage\ 1\ treatment\ effect) + (1 - weight)(stage\ 2\ treatment\ effect)}{\sqrt{(weight)^2(stage\ 1\ standard\ error)^2 + (1 - weight)^2(stage\ 2\ standard\ error)^2}}$$

$$= \frac{.5(-2.29662) + .5(-3.63946)}{\sqrt{(.5)^2(1.59003)^2 + (.5)^2(1.54847)^2}}$$

= -2.67

The Z_{OLS} statistic is then used to calculate the overall OLS p-value. Considering a 2 sided p-value the follow SAS code can be used:

```
Pvalue  = 2*(1-prbnorm(abs(ZOLS));  /*Where ZOLS is the weighted z-statistic.*/
        = 2*(1-probnorm(abs(-2.67));
        = 0.0075
```

SUR METHOD

For the SUR method, similar to the OLS method, Stage 1 (A vs. B from Figure 1) and Stage 2 (placebo non-responders only – C vs.D from Figure 1) are analyzed separately and the weighted treatment effects from the stages are combined to give an overall treatment effect. The SUR method takes into account the fact that random error from the two stages of the study may be correlated for subjects with data in both stages. For subjects who received Drug in Stage 1 or who were placebo responders, any data from Stage 2 are set to missing.

For this method, SAS PROC MODEL was applied for each stage to obtain estimates of the coefficients, using data from that stage to test the null hypothesis that the change due to placebo was equal to that due to Drug versus the alternative that there was a difference between treatments. Note that the SAS/ETS software is needed to run PROC MODEL. In the PROC MODEL statement, the missing=pairwise statement is being used to account for data from all included subjects.

SAS code for this method is given below:

```
*********************************************************;
********************** SUR Method ***********************;
*********************************************************;
/*** Before sending the data into proc model include   ***/
/*** only the data that should be modeled.  In our     ***/
/*** example, all Stage 1 subjects and only Stage 2    ***/
/*** placebo non-responder data should be considered.  ***/
data SPCD2;
       set SPCD;
              if RESPFL = "Y" then do;
                     BASE2 = .; CHG2 = .; TRT02PN =.;
              end;
run;

/*** Model paramaters a0, a1, a2, b0, b1, b2 are not   ***/
/*** defined prior to the model procedure. The         ***/
/*** procedure will define the paramaters during       ***/
/*** model run.                                        ***/
ods output ParameterEstimates=sur Testresults=TR;
ods graphics off;
       proc model data=SPCD2;
              CHG  = a0 + a1 * BASE + a2 * TRT01PN;
              CHG2 = b0 + b1 * BASE2 + b2 * TRT02PN;
                     fit c1 c2/covb sur missing=pairwise;
                     test “Weighted Treatment Effect” .5*a2 + .5*b2 = 0, /wald;
       run;
```

The output produced by this code for our sample data is as follows:

The SAS System

The MODEL Procedure

Nonlinear SUR Summary of Residual Errors

Equation	DF Model	DF Error	SSE	MSE	Root MSE	R-Square	Adj R-Sq
CHG	3	19	184.3	9.7007	3.1146	0.1023	0.0078
CHG2	3	8	46.7420	5.8427	2.4172	0.3374	0.1718

Nonlinear SUR Parameter Estimates

Parameter	Estimate	Approx Std Err	t Value	Approx Pr > |t|
a0	-1.94968	2.3629	-0.83	0.4195
a1	0.24673	0.4236	0.58	0.5671
a2	-2.53949	1.5805	-1.61	0.1246
b0	3.595926	2.4221	1.48	0.1759
b1	-0.62827	0.2847	-2.21	0.0584
b2	-1.95774	1.6386	-1.19	0.2664

Test Results

Test	Type	Statistic	Pr > ChiSq	Label
Weighted Treatment Effect	Wald	3.76	0.0526	.5*a2 + .5*b2 = 0

Average Observations per Equation	16.5000
Average Observations * System Objective	26.8833

Covariances of Parameter Estimates

	a0	a1	a2	b0	b1	b2
a0	5.583296	-.9262898	1.088010	0.022395	-.0028630	0.025912
a1	-0.926290	0.1794018	-0.365362	0.007096	0.0009614	-0.011733
a2	1.088010	-.3653625	2.498117	-0.055677	-.0065735	0.100051
b0	0.022395	0.0070960	-0.055677	5.866627	-.5671118	-2.460373
b1	-0.002863	0.0009614	-0.006574	-0.567112	0.0810542	0.080774
b2	0.025912	-.0117326	0.100051	-2.460373	0.0807736	2.685045

Using the parameterestimates dataset, the Z_{SUR} test statistic can be calculated as:

$$Z_{SUR} = \frac{weight(stage\ 1\ treatment\ effect) + (1 - weight)(stage\ 2\ treatment\ effect)}{\sqrt{(weight)^2(stage\ 1\ standard\ error)^2 + 2(weight)(1 - weight)Cov(stage1\ and\ stage2\ \ treatment\ effect\ ) + (1 - weight)^2(stage\ 2\ standard\ error)^2}}$$

$$= \frac{.5(-2.53949) + .5(-1.95774)}{\sqrt{(.5)^2(1.5805)^2 + 2(.5)(.5)(0.100051) + (.5)^2(1.6386)^2}}$$

$$= -1.9383$$

The Z_{SUR} statistic is then used to calculate the overall SUR p-value. Considering a 2 sided p-value the following SAS code can be used:

```
Pvalue  = 2*(1-prbnorm(abs(ZSUR));  /*Where ZSUR is z-statistic calculated above.*/
            = 2*(1-probnorm(abs(-1.9383);
            = 0.0525821736
```

Note that during the PROC MODEL statement run, a Wald test statistic was requested. The results from this test are stored in the ParameterEstimates(SUR) testresults(TR) datasets and are shown in the SAS output above, The Wald test is based on a chi-square test with one degree of freedom and is equivalent to the Z_{SUR} calculation. The square of the Z_{SUR} statistic is approximately Wald's statistic, (-1.9383)2 = 3.76. Therefore the SAS output can be used directly to obtain the test statistic and p-value.

MMRM METHOD

The MMRM method is a repeated measures linear model that uses all outcome data collected in the trial and accounts for data that were missing at random. The PROC MIXED model determines 3 sets of estimates: one set for Stage 1 (A vs.B), one set for the Stage 2 placebo non-responders (C vs.D) and one set for Stage 2 results for the placebo responders and those who were randomized to Drug in Stage 1 (E vs. F+G). The contrast of interest represents the weighted average treatment effect in all subjects in Stage 1 and in placebo non-responders in Stage 2. Note that each set of estimates could be based on data from additional visits as well, if applicable. In our example, only the Week 5 and Week 10 visits are included. The test statistic is calculated as follows:

$$T = \frac{weight(stage\ 1\ treatment\ effect) + (1 - weight)(stage\ 2\ treatment\ effect)}{\sqrt{(weight)^2(stage\ 1\ standard\ error)^2 + 2(weight)(1 - weight)Cov(stage1\ and\ stage2\ \ treatment\ effect\ ) + (1 - weight)^2(stage\ 2\ standard\ error)^2}}$$

$$= \frac{.5(-2.0043) + .5(-3.7882)}{\sqrt{(.5)^2(1.5590)^2 + 2(.5)(.5)(0.04305) + (.5)^2(1.5604)^2}} = -2.60316$$

```
Pvalue  = 2*(1-probnorm(abs(T));
            = 2*(1-probnorm(abs(-2.60316)));
            = 0.0092
```

Note that in the above equation the parameter estimates are from a model that includes all available data.

SAS code for this method is given below:

```
*********************************************************;
********************** MMRM Method **********************;
*********************************************************;

/*** Create records for each visit                       ***/
proc transpose data=spcd out=spcd2(rename=(_NAME_=AVISIT COL1=AVAL));
      by SUBJID TRT01PN TRT02PN BASE BASE2 RESPFL CHG CHG2;
      var WEEK5 WEEK10;
run;

/*** Create three model flags, set the baseline values  ***/
/*** for each model,a stage flag,and a change variable  ***/
data spcd3;
      set spcd2;
            if AVISIT = "WEEK5" then do;
                  MODEL = 1; STAGE=1;
                  TRT1=TRT01PN; TRT2=0; TRT3=0;
                  BL1=BASE; BL2=0; BL3=0;
                  CHANGE=CHG;
            end;
            if AVISIT = "WEEK10" and (RESPFL = "N" and TRT01PN=0)then do;
                  MODEL = 2; STAGE=2;
                  TRT1=0; TRT2=TRT02PN; TRT3=0;
                  BL1=0; BL2=BASE2; BL3=0;
                  CHANGE=CHG2;
            end;
            if AVISIT = "WEEK10" and (RESPFL = "Y" or (TRT01PN=1 and TRT02PN ne .)) then do;
                  MODEL = 3; STAGE=2;
                  TRT1=0; TRT2=0; TRT3=TRT02PN;
                  BL1=0; BL2=0; BL3=BASE2;
                  CHANGE=CHG2;
            end;

run;
/*** Repeated Measures Model, using an  Unstructured ***/
/*** Covariance Matrix.                              ***/
ods output estimates=est ;
proc mixed data=spcd3 method=reml;
     class STAGE MODEL SUBJID;
     model CHANGE = MODEL BL1 TRT1 BL2 TRT2 BL3 TRT3 /s noint covb;
     repeated STAGE/subject = SUBJID type=un;
     estimate 'Weight' TRT1 0.5 TRT2 0.5;
run;
```

The output produced by this code for our sample data is as follows:

Solution for Fixed Effects						
Effect	MODEL	Estimate	Standard Error	DF	t Value	Pr > |t|
MODEL	1	-2.0225	2.3231	22	-0.87	0.3934
MODEL	2	4.8209	2.3601	22	2.04	0.0532
MODEL	3	1.7760	2.0516	22	0.87	0.3960
BL1		0.2204	0.4157	22	0.53	0.6013
TRT1		-2.0043	1.5590	22	-1.29	0.2119
BL2		-0.8498	0.2907	22	-2.92	0.0079
TRT2		-3.7882	1.5604	22	-2.43	0.0238
BL3		-0.4126	0.2304	22	-1.79	0.0871
TRT3		0.1582	2.2317	22	0.07	0.9441

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > |t|
Weight	-2.8962	1.1126	22	-2.60	0.0162

Covariance Matrix for Fixed Effects										
Row	Effect	MODEL	Col1	Col2	Col3	Col4	Col5	Col6	Col7	Col8	Col9
1	MODEL	1	5.3970	0.1850	0.4963	-0.8921	1.0472	-0.02470	0.1087	-0.04825	-0.1123
2	MODEL	2	0.1850	5.5699	0.05654	0.005817	-0.2226	-0.5903	-1.6738	-0.00554	-0.02269
3	MODEL	3	0.4963	0.05654	4.2090	-0.05449	-0.09933	-0.00151	0.006640	-0.00738	-4.1581
4	BL1		-0.8921	0.005817	-0.05449	0.1728	-0.3533	0.004786	-0.02106	0.005264	0.005423
5	TRT1		1.0472	-0.2226	-0.09933	-0.3533	2.4304	-0.00978	0.04305	0.03560	0.1743
6	BL2		-0.02470	-0.5903	-0.00151	0.004786	-0.00978	0.08449	0.04160	0.000146	0.000150
7	TRT2		0.1087	-1.6738	0.006640	-0.02106	0.04305	0.04160	2.4349	-0.00064	-0.00066
8	BL3		-0.04825	-0.00554	-0.00738	0.005264	0.03560	0.000146	-0.00064	0.05309	-0.1326
9	TRT3		-0.1123	-0.02269	-4.1581	0.005423	0.1743	0.000150	-0.00066	-0.1326	4.9805

The MMRM method uses all the data and is represented in a single model. Even though BL3, TRT3 are not part of the estimate statement, they do contribute to the other parameter estimates. The overall treatment effect, is in the estimates(est) dataset. The test statistic (-2.60316) and p-value (0.009236) are calcualted above.

CONCLUSION

Trials using the SPCD design can be analyzed in several different ways that have been described in the literature. This paper explains how to use the SAS output to construct the test statistic for three different methods: OLS, SUR and MMRM. While the OLS method is the most straightforward, the SUR method takes into account the correlation between subjects who were in both stages, and the MMRM method accounts for data that were missing at random and uses all data in the model.

REFERENCES

1. Fava M., Evins A., Dorer D., Schoenfeld D. The problem of the placebo response in clinical trials for psychiatric disorders: culprits, possible remedies, and a novel study design approach. Psychotherapy and Psychosomatics 2003; 72:115-127; and “Erratum” 2004; 73: 123.
2. Chen, Y, Yang Y., James Hung HM, Wang S-J. Evaluation of performance of some enrichment designs dealing with high placebo response in psychiatric clinical trials. Contemporary Clinical Trials 2011;32:592-604.
3. Tamura R, Huang X. An examination of the efficiency of the sequential parallel design in psychiatric clinical trials. Clin Trials 2007;4:309-17.
4. Doros, G, Pencina M, Rybin D, Meisner A, Fava M. A repeated measures model for analysis of continuous outcomes in sequential parallel comparison design studies. Statistics in Medicine 2013;32:2767-2789.

CONTACT INFORMATION

Harry Haber
MMS Holdings, Inc.
6880 Commerce Blvd.
Canton, MI 48187
Direct: 734-738-5012
FAX: 734-245-0170
hhaber@mmsholdings.com
www.mmsholdings.com

James Zuazo
MMS Holdings, Inc.
6880 Commerce Blvd.
Canton, MI 48187
Direct: 734-245-0460
FAX: 734-245-0170
jzuazo@mmsholdings.com
www.mmsholdings.com

Linda LaMoreaux
MMS Holdings, Inc.
6880 Commerce Blvd.
Canton, MI 48187
Direct: 734- 245-0324
FAX: 734-245-0170
llamoreaux@mmsholdings.com
www.mmsholdings.com